Multivariate Difference Score Models for Drug Acquisition in Rats

Original Source: Lab data from Rick Bevins
Evidence of drug acquisition is operationalized through the *difference* in dipper entry rates for rats when given saline or nicotine. The mean difference scores as a function of nicotine amount are shown below, of which we will use sessions 1, 11, and 22 for this example.



SAS Syntax and Output for Data Manipulation:

```
* Stacking original data (one row per rat) into longitudinal (one row per time per rat);
DATA Rat_byTime; SET Rat_mult;
       ARRAY difscore(22) one--twentytwo;
       ARRAY nicscore(22) nic1--nic22;
       ARRAY salscore(22) sal1--sal22;
       DO i=1 TO 22;
              time=(i); diff=difscore(i)*100; nic=nicscore(i)*100; sal=salscore(i)*100;
              OUTPUT;
       END; DROP i one--sal22;
RUN;
```

Original Repeated Measures ANOVA Model (assuming Compound Symmetry for the Multivariate Normal Distribution and Identity Link): This model predicts the difference score for rat *r* at time *t* →

$$DifScore_{rt} = \beta_0 + \beta_1 Time1_{rt} + \beta_2 Time11_{rt} + \beta_3 Dose1_{rt} + \beta_4 Dose2_{rt} + \beta_5 Time1_{rt} * Dose1_{rt} + \beta_6 Time1_{rt} * Dose2_{rt} + \beta_7 Time11_{rt} * Dose1_{rt} + \beta_8 Time11_{rt} * Dose2_{rt} + e_{rt}$$

```
TITLE1 "Univariate (CS) Repeated Measures ANOVA of Difference Scores";
PROC MIXED DATA=Rat_byTime COVTEST IC NOPROFILE NAMELEN=50 METHOD=ML;
       WHERE time IN(1,11,22);
       CLASS time dose;
       MODEL diff = time|dose / SOLUTION DDFM=KR;
       REPEATED time / R RCORR TYPE=CS SUBJECT=rat; RUN;
```

Class Level Information		
Class	Levels	Values
time	3	1 11 22
dose	3	0.1 0.2 0.4

Iteration History			
Iteration	Evaluations	-2 Log Like	Criterion
0	1	574.49432371	
1	1	573.40760412	0.00000000

Estimated R Matrix for Subject 1			
Row	Col1	Col2	Col3
1	54.6714	6.4145	6.4145
2	6.4145	54.6714	6.4145
3	6.4145	6.4145	54.6714

Estimated R Correlation Matrix for Subject 1			
Row	Col1	Col2	Col3
1	1.0000	0.1173	0.1173
2	0.1173	1.0000	0.1173
3	0.1173	0.1173	1.0000

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
573.4	11	595.4	599.1	599.9	610.1	621.1

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
time	2	56	52.64	<.0001
dose	2	28	0.81	0.4530
time*dose	4	56	0.62	0.6499

Multivariate Repeated Measures ANOVA Model (assuming Unstructured for the Multivariate Normal Distribution and Identity Link): This model predicts the difference score for rat *r* at time *t* →

$$Dipper_{rt} = \beta_0 + \beta_1 Time1_{rt} + \beta_2 Time11_{rt} + \beta_3 Dose1_{rt} + \beta_4 Dose2_{rt} + \beta_5 Time1_{rt} * Dose1_{rt} + \beta_6 Time1_{rt} * Dose2_{rt} + \beta_7 Time11_{rt} * Dose1_{rt} + \beta_8 Time11_{rt} * Dose2_{rt} + e_{rt}$$

```
TITLE1 "Multivariate (UN) Repeated Measures ANOVA of Difference Scores";
PROC MIXED DATA=Rat_byTime COVTEST IC NOPROFILE NAMELEN=50 METHOD=ML;
       WHERE time IN(1,11,22);
       CLASS time dose;
       MODEL diff = time|dose / SOLUTION DDFM=KR;
       REPEATED time / R RCORR TYPE=UN SUBJECT=rat; RUN;
```

Estimated R Matrix for Subject 1			
Row	Col1	Col2	Col3
1	38.7840	-0.4529	-11.7934
2	-0.4529	56.6763	31.4899
3	-11.7934	31.4899	68.5539

Estimated R Correlation Matrix for Subject 1			
Row	Col1	Col2	Col3
1	1.0000	-0.00966	-0.2287
2	-0.00966	1.0000	0.5052
3	-0.2287	0.5052	1.0000

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
562.0	15	592.0	599.1	598.1	612.0	627.0

Does allowing different residual variances and covariance by time significantly improve the model?

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
time	2	27	36.38	<.0001
dose	2	28	0.81	0.4530
time*dose	4	31.6	0.81	0.5261

What if we modeled the difference between saline and nicotine as a model parameter instead? This becomes a doubly multivariate model, which requires stacking the data again:

```
* Stacking longitudinal data into multivariate data (one row per DV per time per rat);
DATA Rat_byDVTime; SET Rat_byTime;
       dipper=nic; dv="Nic"; dvNic=1; dvSal=0; OUTPUT;
       dipper=sal; dv="Sal"; dvNic=0; dvSal=1; OUTPUT;
       * Keeping only times for example;
       WHERE time IN(1,11,22);  RUN;
```

Multivariate Repeated Measures ANOVA Model (assuming Unstructured for the Multivariate Normal Distribution and Identity Link): This model predicts the dipper outcome per DV for rat *r* at time *t* →

$$Dipper_{rt} = \beta_0 + \beta_1 Time1_{rt} + \beta_2 Time11_{rt} + \beta_3 Dose1_{rt} + \beta_4 Dose2_{rt} + \beta_5 Time1_{rt} * Dose1_{rt} + \beta_6 Time1_{rt} * Dose2_{rt} + \beta_7 Time11_{rt} * Dose1_{rt} + \beta_8 Time11_{rt} * Dose2_{rt} + \beta_9 DV_{rt} + \beta_{10} Time1_{rt} * DV_{rt} + \beta_{11} Time11_{rt} * DV_{rt} + \beta_{12} Dose1_{rt} * DV_{rt} + \beta_{13} Dose2_{rt} * DV_{rt} + \beta_{14} Time1_{rt} * Dose1_{rt} * DV_{rt} + \beta_{15} Time1_{rt} * Dose2_{rt} * DV_{rt} + \beta_{16} Time11_{rt} * Dose1_{rt} * DV_{rt} + \beta_{17} Time11_{rt} * Dose2_{rt} * DV_{rt} + e_{rt}$$

```
TITLE1 "Doubly Multivariate Repeated Measures ANOVA";
TITLE2 "Testing Effects for Saline and Diffs for Nicotine";
PROC MIXED DATA=Rat_byDVTime COVTEST IC NOPROFILE NAMELEN=50 METHOD=ML NOINFO;
       CLASS dv time dose;
       MODEL dipper = dv|time|dose / SOLUTION DDFM=KR;
       REPEATED dv*time / R RCORR TYPE=UN SUBJECT=rat;
       LSMEANS dv*time dv*dose /
         SLICE=dv;          /* Test marginal effects of time (df=2), dose (df=2) per dv */
         *DIFF=ALL;         /* Get all possible marginal mean differences */
       LSMEANS time*dose time*dv /
         SLICE=time;        /* Test marginal effects of dose (df=2), dv (df=1) per time */
         *DIFF=ALL;         /* Get all possible marginal mean differences */
       LSMEANS dose*time dose*dv /
         SLICE=dose;        /* Test marginal effects of time (df=2), dv (df=1) per dose */
         *DIFF=ALL;         /* Get all possible marginal mean differences */
       LSMEANS dv*time*dose /
         SLICE=dv*time      /* Test marginal effect of dose (df=2) per dv and time */
         SLICE=dv*dose      /* Test marginal effect of time (df=2) per dv and dose */
         SLICE=time*dose;   /* Test marginal effect of dv (df=1) dv per time and dose */
         *DIFF=ALL;         /* Get all possible cell mean differences */
RUN;
```

Class Level Information

Class	Levels	Values
dv	2	Nic Sal
time	3	1 11 22
dose	3	0.1 0.2 0.4

Iteration History

Iteration	Evaluations	-2 Log Like	Criterion
0	1	1037.92286025	
1	1	909.68224381	0.00000000

Estimated R Matrix for Subject 1

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	3.0320	-3.4788	0.8587	-2.6338	-3.5572	0.1579
2	-3.4788	28.7944	14.2243	10.2788	10.7932	2.7149
3	0.8587	14.2243	60.2217	8.4229	38.6018	-3.8793
4	-2.6338	10.2788	8.4229	13.3005	12.2913	1.3001
5	-3.5572	10.7932	38.6018	12.2913	55.9179	-2.2740
6	0.1579	2.7149	-3.8793	1.3001	-2.2740	8.0881

Information Criteria

Neg 2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
909.7	39	987.7	1012.1	1003.6	1039.6	1078.6

Estimated R Correlation Matrix for Subject 1

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	-0.3723	0.06355	-0.4148	-0.2732	0.03189
2	-0.3723	1.0000	0.3416	0.5252	0.2690	0.1779
3	0.06355	0.3416	1.0000	0.2976	0.6652	-0.1758
4	-0.4148	0.5252	0.2976	1.0000	0.4507	0.1253
5	-0.2732	0.2690	0.6652	0.4507	1.0000	-0.1069
6	0.03189	0.1779	-0.1758	0.1253	-0.1069	1.0000

Solution for Fixed Effects

Effect	dv	time	dose	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept				4.4750	1.0055	28	4.45	0.0001
dv	Nic			15.4462	2.9273	28	5.28	<.0001
dv	Sal			0	.	.	.	.
time		1		1.4188	1.9828	28	0.72	0.4802
time		11		4.2463	1.5325	28	2.77	0.0098
time		22		0	.	.	.	.
dv*time	Nic	1		-19.2563	4.0454	28	-4.76	<.0001
dv*time	Nic	11		-6.8250	2.7895	28	-2.45	0.0210
dv*time	Nic	22		0	.	.	.	.
dv*time	Sal	1		0	.	.	.	.
dv*time	Sal	11		0	.	.	.	.
dv*time	Sal	22		0	.	.	.	.
dose			0.1	-0.5210	1.3490	28	-0.39	0.7023
dose			0.2	-0.4440	1.3490	28	-0.33	0.7445
dose			0.4	0	.	.	.	.
dv*dose	Nic		0.1	-4.7912	3.9274	28	-1.22	0.2327
dv*dose	Nic		0.2	-0.6472	3.9274	28	-0.16	0.8703
dv*dose	Nic		0.4	0	.	.	.	.
dv*dose	Sal		0.1	0	.	.	.	.
dv*dose	Sal		0.2	0	.	.	.	.
dv*dose	Sal		0.4	0	.	.	.	.
time*dose		1	0.1	1.9562	2.6602	28	0.74	0.4682
time*dose		1	0.2	1.8492	2.6602	28	0.70	0.4927
time*dose		1	0.4	0	.	.	.	.
time*dose		11	0.1	-0.5268	2.0561	28	-0.26	0.7997
time*dose		11	0.2	-2.0803	2.0561	28	-1.01	0.3203
time*dose		11	0.4	0	.	.	.	.
time*dose		22	0.1	0	.	.	.	.
time*dose		22	0.2	0	.	.	.	.
time*dose		22	0.4	0	.	.	.	.
dv*time*dose	Nic	1	0.1	3.7333	5.4275	28	0.69	0.4972
dv*time*dose	Nic	1	0.2	0.04625	5.4275	28	0.01	0.9933
dv*time*dose	Nic	1	0.4	0	.	.	.	.
dv*time*dose	Nic	11	0.1	5.6125	3.7425	28	1.50	0.1449
dv*time*dose	Nic	11	0.2	4.9200	3.7425	28	1.31	0.1993
dv*time*dose	Nic	11	0.4	0	.	.	.	.
dv*time*dose	Nic	22	0.1	0	.	.	.	.
dv*time*dose	Nic	22	0.2	0	.	.	.	.
dv*time*dose	Nic	22	0.4	0	.	.	.	.
dv*time*dose	Sal	1	0.1	0	.	.	.	.
dv*time*dose	Sal	1	0.2	0	.	.	.	.
dv*time*dose	Sal	1	0.4	0	.	.	.	.
dv*time*dose	Sal	11	0.1	0	.	.	.	.

What are the reference DV, time, and dose, as determined by the CLASS statement?

There are 18 non-zero estimates in this table that correspond to the 18 β fixed effects in the model equation.

All of these effects can be summarized by tests of marginal main effects and interactions, as seen below, but more work will be needed to interpret the simple effects embedded within these.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
dv	1	All8	52.48	<.0001
time	2	27	53.01	<.0001
dv*time	2	27	36.38	<.0001
dose	2	28	0.26	0.7715
dv*dose	2	28	0.81	0.4530
time*dose	4	31.6	1.13	0.3603
dv*time*dose	4	31.6	0.81	0.5261

The solution for fixed effects keeps going with rows of 0s, but it is truncated here to save space.

Least Squares Means									
Effect	dv	time	dose	Estimate	Standard Error	DF	t Value	Pr > |t|	Where this stuff came from:
dv*time	Nic	1		2.4776	0.3309	28	7.49	<.0001	LSMEANS dv*time dv*dose / SLICE=dv;
dv*time	Nic	11		17.8498	1.4747	28	12.10	<.0001	
dv*time	Nic	22		17.7868	1.4210	28	12.52	<.0001	
dv*time	Sal	1		6.8406	1.0197	28	6.71	<.0001	
dv*time	Sal	11		7.5306	0.6930	28	10.87	<.0001	
dv*time	Sal	22		4.1533	0.5404	28	7.69	<.0001	
dv*dose	Nic		0.1	11.3953	1.4567	28	7.82	<.0001	
dv*dose	Nic		0.2	13.6030	1.4567	28	9.34	<.0001	
dv*dose	Nic		0.4	13.1158	1.6286	28	8.05	<.0001	
dv*dose	Sal		0.1	6.3188	0.9355	28	6.75	<.0001	
dv*dose	Sal		0.2	5.8423	0.9355	28	6.24	<.0001	
dv*dose	Sal		0.4	6.3633	1.0460	28	6.08	<.0001	
time*dose		1	0.1	4.8950	0.7885	28	6.21	<.0001	LSMEANS time*dose time*dv / SLICE=time;
time*dose		1	0.2	5.0935	0.7885	28	6.46	<.0001	
time*dose		1	0.4	3.9888	0.8816	28	4.52	0.0001	
time*dose		11	0.1	12.3948	1.5031	28	8.25	<.0001	
time*dose		11	0.2	12.6440	1.5031	28	8.41	<.0001	
time*dose		11	0.4	13.0319	1.6805	28	7.75	<.0001	
time*dose		22	0.1	9.2815	1.2192	28	7.61	<.0001	
time*dose		22	0.2	11.4305	1.2192	28	9.38	<.0001	
time*dose		22	0.4	12.1981	1.3631	28	8.95	<.0001	
dv*time*dose	Nic	1	0.1	2.4610	0.5506	28	4.47	0.0001	
dv*time*dose	Nic	1	0.2	2.8880	0.5506	28	5.24	<.0001	LSMEANS dv*time*dose / SLICE=dv*time SLICE=dv*dose SLICE=time*dose;
dv*time*dose	Nic	1	0.4	2.0837	0.6156	28	3.38	0.0021	
dv*time*dose	Nic	11	0.1	17.1160	2.4540	28	6.97	<.0001	
dv*time*dose	Nic	11	0.2	19.0910	2.4540	28	7.78	<.0001	
dv*time*dose	Nic	11	0.4	17.3425	2.7437	28	6.32	<.0001	
dv*time*dose	Nic	22	0.1	14.6090	2.3647	28	6.18	<.0001	
dv*time*dose	Nic	22	0.2	18.8300	2.3647	28	7.96	<.0001	
dv*time*dose	Nic	22	0.4	19.9212	2.6438	28	7.54	<.0001	
dv*time*dose	Sal	1	0.1	7.3290	1.6969	28	4.32	0.0002	
dv*time*dose	Sal	1	0.2	7.2990	1.6969	28	4.30	0.0002	
dv*time*dose	Sal	1	0.4	5.8938	1.8972	28	3.11	0.0043	
dv*time*dose	Sal	11	0.1	7.6735	1.1533	28	6.65	<.0001	
dv*time*dose	Sal	11	0.2	6.1970	1.1533	28	5.37	<.0001	
dv*time*dose	Sal	11	0.4	8.7213	1.2894	28	6.76	<.0001	
dv*time*dose	Sal	22	0.1	3.9540	0.8993	28	4.40	0.0001	
dv*time*dose	Sal	22	0.2	4.0310	0.8993	28	4.48	0.0001	
dv*time*dose	Sal	22	0.4	4.4750	1.0055	28	4.45	0.0001	

Tests of Effect Slices								
Effect	dv	time	dose	Num DF	Den DF	F Value	Pr > F	What this is testing:
dv*time	Nic			2	27	58.29	<.0001	Effects of time per dv (averaged over dose)
dv*time	Sal			2	27	8.11	0.0017	
dv*dose	Nic			2	28	0.63	0.5424	Effects of dose per dv (averaged over time)
dv*dose	Sal			2	28	0.09	0.9133	
time*dose		1		2	28	0.48	0.6235	Effects of dose per time (averaged over dv)
time*dose		11		2	28	0.04	0.9608	
time*dose		22		2	28	1.43	0.2565	
dv*time		1		1	28	13.59	0.0010	Effects of dv per time (averaged over dose)
dv*time		11		1	28	52.03	<.0001	
dv*time		22		1	28	75.08	<.0001	
time*dose			0.1	2	27	15.55	<.0001	Effects of time per dose (averaged over dv)
time*dose			0.2	2	27	17.59	<.0001	
time*dose			0.4	2	27	21.44	<.0001	
dv*dose			0.1	1	28	11.45	0.0021	Effects of dv per dose (averaged over time)
dv*dose			0.2	1	28	26.77	<.0001	
dv*dose			0.4	1	28	16.21	0.0004	
dv*time*dose	Nic	1		2	28	0.48	0.6243	Effects of dose per dv and time
dv*time*dose	Nic	11		2	28	0.19	0.8285	
dv*time*dose	Nic	22		2	28	1.32	0.2832	
dv*time*dose	Sal	1		2	28	0.20	0.8197	
dv*time*dose	Sal	11		2	28	1.09	0.3484	
dv*time*dose	Sal	22		2	28	0.08	0.9196	
dv*time*dose	Nic		0.1	2	27	17.18	<.0001	Effects of time per dv and time
dv*time*dose	Nic		0.2	2	27	23.14	<.0001	
dv*time*dose	Nic		0.4	2	27	19.89	<.0001	
dv*time*dose	Sal		0.1	2	27	3.62	0.0406	
dv*time*dose	Sal		0.2	2	27	1.82	0.1819	
dv*time*dose	Sal		0.4	2	27	4.33	0.0234	
dv*time*dose		1	0.1	1	28	6.11	0.0198	Effects of dv per time and dose
dv*time*dose		1	0.2	1	28	5.02	0.0332	
dv*time*dose		1	0.4	1	28	2.99	0.0946	
dv*time*dose		11	0.1	1	28	15.73	0.0005	
dv*time*dose		11	0.2	1	28	29.33	<.0001	
dv*time*dose		11	0.4	1	28	10.49	0.0031	
dv*time*dose		22	0.1	1	28	16.56	0.0003	
dv*time*dose		22	0.2	1	28	31.95	<.0001	
dv*time*dose		22	0.4	1	28	27.84	<.0001	